Hopf bifurcation cannot occur for the (RDB) system with equal diffusion coefficients

Definition of matrix of the linearised system

```
J = {{Fu, fv, fw}, {gu, gv, 0}, {hu, 0, hw}} /. Fu → (fu - hu);
```

and of the matrix of diffusion coefficients

```
Ds = {{du, 0, 0}, {0, dv, 0}, {0, 0, 0}};
```

Routh-Hurwitz conditions

```
coef0 = Coefficient[-Det[J - λ IdentityMatrix[3]], λ, 0];
coef1 = Coefficient[-Det[J - λ IdentityMatrix[3]], λ, 1];
coef2 = Coefficient[-Det[J - λ IdentityMatrix[3]], λ, 2];
coef3 = Coefficient[-Det[J - λ IdentityMatrix[3]], λ, 3];
RH = FullSimplify[
  coef0 > 0 && coef1 > 0 && coef2 > 0 && coef3 > 0 && coef2 coef1 > coef3 coef0]
```

$f_v g_u h_w + g_v (-f_u h_w + h_u (f_w + h_w)) > 0$ && $g_v h_w + f_u (g_v + h_w) > f_v g_u + h_u (f_w + g_v + h_w)$ && $h_u > f_u + g_v + h_w$ && $f_w g_v h_u + (f_v g_u + g_v (-f_u + h_u)) h_w + (f_u + g_v - h_u + h_w)(-f_v g_u - (f_w + g_v) h_u + (g_v - h_u) h_w + f_u (g_v + h_w)) < 0$

Routh-Hurwitz conditions but in another format (commas instead of logical and &&)

```
RHcomma = Table[RH[[j]], {j, 1, Length[RH]}]
```

$\{f_v g_u h_w + g_v (-f_u h_w + h_u (f_w + h_w)) > 0,\ g_v h_w + f_u (g_v + h_w) > f_v g_u + h_u (f_w + g_v + h_w),\ h_u > f_u + g_v + h_w,\ f_w g_v h_u + (f_v g_u + g_v (-f_u + h_u)) h_w + (f_u + g_v - h_u + h_w)(-f_v g_u - (f_w + g_v) h_u + (g_v - h_u) h_w + f_u (g_v + h_w)) < 0\}$

dispersion relation, C(*κ*,p) and solution of C(*κ*,p)

```
dispRel = Det[ω IdentityMatrix[3] + κ Ds - J];
Cκp = Coefficient[dispRel, ω, 0];
κBif = κ /. Solve[Cκp == 0, κ]
```

$$\left\{\frac{1}{2 d_u d_v h_w}\left(-d_v f_w h_u + d_v f_u h_w + d_u g_v h_w - d_v h_u h_w - \sqrt{\left((-d_v f_w h_u + d_v f_u h_w + d_u g_v h_w - d_v h_u h_w)^2 + 4 d_u d_v h_w (f_w g_v h_u + f_v g_u h_w - f_u g_v h_w + g_v h_u h_w)\right)}\right), \frac{1}{2 d_u d_v h_w}\left(-d_v f_w h_u + d_v f_u h_w + d_u g_v h_w - d_v h_u h_w + \sqrt{\left((-d_v f_w h_u + d_v f_u h_w + d_u g_v h_w - d_v h_u h_w)^2 + 4 d_u d_v h_w (f_w g_v h_u + f_v g_u h_w - f_u g_v h_w + g_v h_u h_w)\right)}\right)\right\}$$

dispersion relation

```
polyλ = dispRel /. ω → λ
```

$-(\lambda f_w + \kappa d_v f_w - f_w g_v) h_u + (-f_v g_u + (\lambda + \kappa d_v - g_v)(\lambda + \kappa d_u - f_u + h_u))(\lambda - h_w)$

A cubic polynomial with real coeffs has always at least one real root. If other root is complex, then its complex conjugate is the last root.

Let us denote λ_R the real root and the complex conjugate pair as $\mu \pm i\,\nu$

```
Expand[polyλ /. λ → (μ + ⅈ ν)] - Expand[polyλ /. λ → (μ - ⅈ ν)];
Eq4μνk = Collect[Simplify[% (-ⅈ)/(2 ν)], {μ, k}]
```

$$3\mu^2 - \nu^2 + \kappa^2 d_u d_v - \kappa d_v f_u - f_v g_u - \kappa d_u g_v + f_u g_v + \kappa d_v h_u - f_w h_u - g_v h_u + \mu\,(2\kappa d_u + 2\kappa d_v - 2 f_u - 2 g_v + 2 h_u - 2 h_w) - \kappa d_u h_w - \kappa d_v h_w + f_u h_w + g_v h_w - h_u h_w$$

By subtraction the degree of the polynomial polyλ was decreased, but now is dependent on the unknown imaginary part $\nu\neq 0$.
Using Vieta's formulas we can obtaion $\nu=\nu(\mu,\kappa)$.

```
ForVieta = Collect[Expand[(λ - λR) (λ - (μ + ⅈ ν)) (λ - (μ - ⅈ ν))], λ]
```

$$\lambda^3 + \lambda^2\,(-2\mu - \lambda_R) - \mu^2\lambda_R - \nu^2\lambda_R + \lambda\left(\mu^2 + \nu^2 + 2\mu\lambda_R\right)$$

```
vieta1 = Coefficient[polyλ, λ^2] == Coefficient[ForVieta, λ^2]
vieta2 = Coefficient[polyλ, λ^1] == Coefficient[ForVieta, λ^1]
vieta3 = Coefficient[polyλ, λ, 0] == Coefficient[ForVieta, λ, 0]
```

$$\kappa d_u + \kappa d_v - f_u - g_v + h_u - h_w == -2\mu - \lambda_R$$

$$\kappa^2 d_u d_v - \kappa d_v f_u - f_v g_u - \kappa d_u g_v + f_u g_v + \kappa d_v h_u - f_w h_u - g_v h_u - \kappa d_u h_w - \kappa d_v h_w + f_u h_w + g_v h_w - h_u h_w == \mu^2 + \nu^2 + 2\mu\lambda_R$$

$$-\kappa d_v f_w h_u + f_w g_v h_u - \kappa^2 d_u d_v h_w + \kappa d_v f_u h_w + f_v g_u h_w + \kappa d_u g_v h_w - f_u g_v h_w - \kappa d_v h_u h_w + g_v h_u h_w == -\mu^2\lambda_R - \nu^2\lambda_R$$

```
RealRoot = Solve[vieta1, λR][[1]]
```

$$\{\lambda_R \to -2\mu - \kappa d_u - \kappa d_v + f_u + g_v - h_u + h_w\}$$

We can see that the real root has to be negative $\lambda_R < 0$ once a Hopf instability occurs as μ>0, $\kappa = k^2 > 0$

```
ImagPart1stExpression = (Solve[vieta2 /. RealRoot /. ν → √nu, nu] /. nu → ν^2)[[1]]
```

$$\{\nu^2 \to 3\mu^2 + 2\kappa\mu d_u + 2\kappa\mu d_v + \kappa^2 d_u d_v - 2\mu f_u - \kappa d_v f_u - f_v g_u - 2\mu g_v - \kappa d_u g_v + f_u g_v + 2\mu h_u + \kappa d_v h_u - f_w h_u - g_v h_u - 2\mu h_w - \kappa d_u h_w - \kappa d_v h_w + f_u h_w + g_v h_w - h_u h_w\}$$

```
Simplify[Collect[Eq4μνk /. ImagPart1stExpression, μ]]
```

0

Thus this combination does not gain anything new, have to use another Vieta's relation

```
ImagPart2ndExpression = (Solve[vieta3 /. RealRoot /. ν → √nu, nu] /. nu → ν^2)[[1]]
```

$$\Big\{\nu^2 \to \frac{1}{2\mu + \kappa d_u + \kappa d_v - f_u - g_v + h_u - h_w}\big(-2\mu^3 - \kappa\mu^2 d_u - \kappa\mu^2 d_v + \mu^2 f_u + \mu^2 g_v - \mu^2 h_u - \kappa d_v f_w h_u + f_w g_v h_u + \mu^2 h_w - \kappa^2 d_u d_v h_w + \kappa d_v f_u h_w + f_v g_u h_w + \kappa d_u g_v h_w - f_u g_v h_w - \kappa d_v h_u h_w + g_v h_u h_w\big)\Big\}$$

```
Eq4μk = Collect[Cancel[(Eq4μνk /. ImagPart2ndExpression)
    Denominator[ν² /. ImagPart2ndExpression]] /. κ → k², {μ, k}, Simplify]
```

$$\begin{aligned}
&8\mu^3 + k^6 d_u d_v (d_u + d_v) + f_v g_u (g_v - h_u) - \\
&f_u^2 (g_v + h_w) + f_u \left(f_v g_u - g_v^2 + f_w h_u + 2 g_v (h_u - h_w) + 2 h_u h_w - h_w^2\right) + \\
&(h_u - h_w)\left(g_v^2 - h_u (f_w + h_w) + g_v (-h_u + h_w)\right) + \mu^2 \left(8 k^2 (d_u + d_v) - 8 (f_u + g_v - h_u + h_w)\right) + \\
&k^4 \left(-d_u^2 (g_v + h_w) - d_v^2 (f_u - h_u + h_w) - 2 d_u d_v (f_u + g_v - h_u + h_w)\right) + \\
&k^2 \left(d_u \left(-f_v g_u + g_v^2 - f_w h_u - 2 g_v h_u + 2 g_v h_w - 2 h_u h_w + h_w^2 + 2 f_u (g_v + h_w)\right) + d_v\right. \\
&\quad \left.\left(f_u^2 - f_v g_u + (h_u - h_w)(-2 g_v + h_u - h_w) + 2 f_u (g_v - h_u + h_w)\right)\right) + \mu \left(2 k^4 \left(d_u^2 + 3 d_u d_v + d_v^2\right) + \right. \\
&\quad 2 \left(f_u^2 - f_v g_u + g_v^2 - f_w h_u - 3 g_v h_u + h_u^2 + 3 g_v h_w - 3 h_u h_w + h_w^2 + f_u (3 g_v - 2 h_u + 3 h_w)\right) - \\
&\quad \left. 2 k^2 \left(d_v (3 f_u + 2 g_v - 3 h_u + 3 h_w) + d_u (2 f_u + 3 g_v - 2 h_u + 3 h_w)\right)\right)
\end{aligned}$$

Therefore we have a polynomial that implicitly defines $\mu = \mu(k^2)$. This polynomial is again cubic.

Consider equal diffusion coefficients. We can take advantage of the following observation: Hopf bifurcation occurs when $\mu \in$Reals crosses zero to positive values; from Vieta's relations we know that this happens when the zeroth term of polynomial Eq4μk vanishes, i.e. Eq4μk(μ=0)=0.

```
neccessaryCond = Collect[Coefficient[Eq4μk, μ, 0], k] /. du → dv
```

$$\begin{aligned}
&2 k^6 d_v^3 + f_v g_u (g_v - h_u) - f_u^2 (g_v + h_w) + f_u \left(f_v g_u - g_v^2 + f_w h_u + 2 g_v (h_u - h_w) + 2 h_u h_w - h_w^2\right) + \\
&(h_u - h_w)\left(g_v^2 - h_u (f_w + h_w) + g_v (-h_u + h_w)\right) + \\
&k^4 \left(-d_v^2 (g_v + h_w) - d_v^2 (f_u - h_u + h_w) - 2 d_v^2 (f_u + g_v - h_u + h_w)\right) + \\
&k^2 \left(d_v \left(-f_v g_u + g_v^2 - f_w h_u - 2 g_v h_u + 2 g_v h_w - 2 h_u h_w + h_w^2 + 2 f_u (g_v + h_w)\right) + \right. \\
&\quad \left. d_v \left(f_u^2 - f_v g_u + (h_u - h_w)(-2 g_v + h_u - h_w) + 2 f_u (g_v - h_u + h_w)\right)\right)
\end{aligned}$$

Binding self-inhibitor

wanted conditions for linearised system (binding self-inhibitor $f_u < 0$)

```
wantedConditions = {du == dv, dv > 0, du > 0, fu < 0,
  gv ∈ Reals, hu > 0, hw < 0, fw > 0, fw == -hw, fv ∈ Reals, gu ∈ Reals}
```

$\{d_u == d_v,\ d_v > 0,\ d_u > 0,\ f_u < 0,\ g_v \in \text{Reals},$
$h_u > 0,\ h_w < 0,\ f_w > 0,\ f_w == -h_w,\ f_v \in \text{Reals},\ g_u \in \text{Reals}\}$

Neccessary condition is again a cubic polynomial in $\kappa = k^2$.

Notice, however, the signs of coefficients in this polynomial:
1. by $\kappa^3 = k^6$ we have 2 $d_v{}^3 > 0$

```
Coefficient[neccessaryCond, k^6]
Reduce[{Coefficient[neccessaryCond, k^6] < 0} ⋃ wantedConditions ⋃ RHcomma]
```

$2 d_v^3$

```
False
```

2. Can the coefficient by $\kappa^2 = k^4$ be negative?

```
Coefficient[neccessaryCond, k^4]
Reduce[
 {Coefficient[neccessaryCond, k^4] < 0} ⋃ wantedConditions ⋃ RHcomma /. d_u → d_v]
```

$-d_v^2 (g_v + h_w) - d_v^2 (f_u - h_u + h_w) - 2 d_v^2 (f_u + g_v - h_u + h_w)$

False

3. Can the coefficient by $\kappa = k^2$ be negative?

```
Coefficient[neccessaryCond, k^2]
Reduce[Reduce[wantedConditions ⋃ (RHcomma /. d_u → d_v)] &&
  Coefficient[neccessaryCond, k^2] < 0]
```

$d_v \left(-f_v g_u + g_v^2 - f_w h_u - 2 g_v h_u + 2 g_v h_w - 2 h_u h_w + h_w^2 + 2 f_u (g_v + h_w)\right) + d_v \left(f_u^2 - f_v g_u + (h_u - h_w) (-2 g_v + h_u - h_w) + 2 f_u (g_v - h_u + h_w)\right)$

False

4. Can the coefficient by $\kappa^0 = k^0$ be negative?

```
Coefficient[neccessaryCond, k, 0]
Reduce[{Coefficient[neccessaryCond, k, 0] < 0} ⋃ wantedConditions ⋃ RHcomma]
```

$f_v g_u (g_v - h_u) - f_u^2 (g_v + h_w) + f_u \left(f_v g_u - g_v^2 + f_w h_u + 2 g_v (h_u - h_w) + 2 h_u h_w - h_w^2\right) + (h_u - h_w) \left(g_v^2 - h_u (f_w + h_w) + g_v (-h_u + h_w)\right)$

False

Therefore, due to the Descarte's rule of signs, we cannot have a positive real root $\kappa = k^2$ of the neccessaryCond.

Binding self-activator

wanted conditions for linearised system (binding self-activator $f_u > 0$)

```
wantedConditions = {d_u == d_v, d_v > 0, d_u > 0, f_u > 0,
  g_v ∈ Reals, h_u > 0, h_w < 0, f_w > 0, f_w == -h_w, f_v ∈ Reals, g_u ∈ Reals}
```

$\{d_u == d_v,\ d_v > 0,\ d_u > 0,\ f_u > 0,\ g_v \in \text{Reals},\ h_u > 0,\ h_w < 0,\ f_w > 0,\ f_w == -h_w,\ f_v \in \text{Reals},\ g_u \in \text{Reals}\}$

Neccessary condition is again a cubic polynomial in $\kappa = k^2$.

Notice, however, the signs of coefficients in this polynomial:
1. by $\kappa^3 = k^6$ we have $2\, d_v{}^3 > 0$

```
Coefficient[neccessaryCond, k^6]
Reduce[{Coefficient[neccessaryCond, k^6] < 0} ⋃ wantedConditions ⋃ RHcomma]
```

$2\, d_v^3$

False

2. Can the coefficient by $\kappa^2 = k^4$ be negative?

```
Coefficient[neccessaryCond, k^4]
Reduce[
 {Coefficient[neccessaryCond, k^4] < 0} ⋃ wantedConditions ⋃ RHcomma /. d_u → d_v]
```

$-d_v^2\,(g_v+h_w)-d_v^2\,(f_u-h_u+h_w)-2\,d_v^2\,(f_u+g_v-h_u+h_w)$

```
False
```

3. Can the coefficient by $\kappa = k^2$ be negative?

```
Coefficient[neccessaryCond, k^2]
Reduce[Reduce[wantedConditions ⋃ (RHcomma /. d_u → d_v)] &&
  Coefficient[neccessaryCond, k^2] < 0]
```

$d_v\,\left(-f_v\,g_u+g_v^2-f_w\,h_u-2\,g_v\,h_u+2\,g_v\,h_w-2\,h_u\,h_w+h_w^2+2\,f_u\,(g_v+h_w)\right)+$
$d_v\,\left(f_u^2-f_v\,g_u+(h_u-h_w)\,(-2\,g_v+h_u-h_w)+2\,f_u\,(g_v-h_u+h_w)\right)$

```
False
```

4. Can the coefficient by $\kappa^0 = k^0$ be negative?

```
Coefficient[neccessaryCond, k, 0]
Reduce[{Coefficient[neccessaryCond, k, 0] < 0} ⋃ wantedConditions ⋃ RHcomma]
```

$f_v\,g_u\,(g_v-h_u)-f_u^2\,(g_v+h_w)+f_u\,\left(f_v\,g_u-g_v^2+f_w\,h_u+2\,g_v\,(h_u-h_w)+2\,h_u\,h_w-h_w^2\right)+$
$(h_u-h_w)\,\left(g_v^2-h_u\,(f_w+h_w)+g_v\,(-h_u+h_w)\right)$

```
False
```

Therefore, due to the Descarte’s rule of signs, we cannot have a positive real root $\kappa = k^2$ of the neccessaryCond.